Paper 091-2008

Creating Journal-Style Tables in an Easy Way (with PROC TABULATE, PROC TEMPLATE, PROC FORMAT and ODS RTF)

Janet Grubber, Maren Olsen, Hayden Bosworth
Durham Veterans Affairs Medical Center/Duke University Medical Center

ABSTRACT

When you are creating "Table 1 – Demographics" type tables for journal articles or presentations, do you find it tedious to manually type or cut and paste output from SAS PROC FREQ or PROC UNIVARIATE into Microsoft Word documents? Do you find yourself spending a lot of time double-checking to make sure you didn't make any typos during that process? Do you feel like you're wasting precious time when you have to go back and manually redo a table because you discover that you should have presented your data stratified by an additional variable (such as race) rather than just your overall data? Fear not…there is a way to cut down on your effort and the need to check for typos with a little bit of up-front effort in SAS.

This paper describes methods for automating the production of demographic-type tables by using PROC TEMPLATE, PROC FORMAT, PROC TABULATE, LABEL statements, and ODS RTF in SAS.

In addition, this paper describes how the following special features can be used in SAS to make the demographic-type tables publication-ready:

- Indenting response levels (e.g. Male, Female) under broader category row headings (i.e. Gender) using the CLASSLEV [ASIS=ON] option along with a PROC FORMAT "trick"
- Creating an "overall" variable that can be used to generate statistics for overall data sets in PROC TABULATE
- Formatting percentages using PICTURE PCTFMT
- Using the COLPCTN statement to present column percentages
- Representing missing text with 0s using the MISSTEXT='0' statement
- Customizing fonts using PROC TEMPLATE

INTRODUCTION

Tables such as Tables 1 and 2 below can be produced using SAS methods that are easily implemented. This paper 1) describes specific steps needed to create these types of tables 2) presents various examples of frequency and means tables paired with the PROC TABULATE code that produced them and 3) provides complete SAS code in an Appendix to show how the separate pieces can be integrated to produce publication ready tables. Special SAS coding features that are helpful in creating these tables are highlighted in yellow throughout.

Table 1. Race and Blood Pressure, Baseline Data
Frequencies, Overall and by BP Control Status

			BP Control			
	Overall		In Control (< 140/90)		Out of Control (>=140/90)	
	N	(%)	N	(%)	N	(%)
Race						
Black	306	(50.3)	156	(44.1)	150	(59.1)
White	302	(49.7)	198	(55.9)	104	(40.9)
Gender						
Female	404	(66.4)	231	(65.3)	173	(68.1)
Male	204	(33.6)	123	(34.7)	81	(31.9)
Married						
Missing	2	(0.3)	1	(0.3)	1	(0.4)
No	300	(49.3)	166	(46.9)	134	(52.8)
Yes	306	(50.3)	187	(52.8)	119	(46.9)
Age						
25-50	120	(19.7)	74	(20.9)	46	(18.1)
51-60	172	(28.3)	106	(29.9)	66	(26.0)
61-70	166	(27.3)	101	(28.5)	65	(25.6)
>70	148	(24.3)	72	(20.3)	76	(29.9)
Missing	2	(0.3)	1	(0.3)	1	(0.4)

Table 2. Race and Blood Pressure, Baseline Data
Systolic Blood Pressure Means, Overall and by Race

						Race									
	Overall					Black					White				
	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max
Clinic Systolic BP	608	135.3	20.2	90.0	226.0	306	139.5	21.9	90.0	226.0	302	131.1	17.4	90.0	186.0
Clinic Diastolic BP	608	77.8	11.1	40.0	120.0	306	80.1	11.7	52.0	120.0	302	75.5	10.0	40.0	102.0
Age	606	61.3	12.3	25.0	92.0	305	59.3	12.5	28.0	92.0	301	63.4	11.7	25.0	87.0

STEPS TO CREATE TABLES SUCH AS TABLES 1 AND 2 (SHOWN ABOVE)

1. **Use PROC FORMAT to assign names to the response levels of categorical variables that you want to use in the tables.**
   a. Insert a fixed number of spaces in front of the names of your response levels. These spaces will be used to indent the response levels (such as Female, Male) under their broader category headings (such as Gender) when you specify ASIS=ON in a CLASSLEV statement within PROC TABULATE (See Table 5a). (I have added 4 spaces for indentation).
   b. If you would like a column in your table for your overall data set, add a format for 'overall' (see below), and be sure to create a corresponding variable (i.e. OVERALL=1) in your data set in a data step (See Step 3). This 'overall' variable will serve as an indicator variable for all data in your data set.
   c. Include the PICTURE PCTFMT(ROUND)LOW-HIGH format in your format statement. This will allow you to represent the percentages in your table in a "pretty" format when you invoke that format in PROC TABULATE code (See Table 5a).

```
proc format;

value yesno_ 1='    Yes'
             0='    No'
             .='    Missing';

value male_  1='    Male'
             0='    Female';

value agecat_ 1='    25-50'
              2='    51-60'
              3='    61-70'
              4='    >70'
              .='    Missing';

value black_  1='    Black'
              0='    White';

value clbpoutc_ 1='    Out of Control (>=140/90)'
                0='    In Control (< 140/90)';

value overall 1='Overall';


picture pctfmt(round)low-high= ' 009.9)' (prefix='(');

run;
```

Note: The yellow-highlighted blank spaces are necessary for proper indention of response levels in tables.

2. **Use the following PROC TEMPLATE code to create a 'journal style' template. This is a slight modification to the 'canned' SAS minimal styles template.**

```
proc template;
  define style styles.janettables;
    parent=styles.minimal;

    style bodyDate from bodyDate /
           font=('Times',8pt);
    style PagesDate from PagesDate /
           font=('Times',8pt);
    style PagesTitle from PagesTitle /
           font=('Times',8pt);
    style SystemTitle from SystemTitle /
           font=('Times',12pt);
    style Data from Data /
          font=('Times, Times, Times',8pt);
    style Header from HeadersAndFooters /
           font=('Times, Times, Times',8pt);
    style RowHeader from HeadersAndFooters /
           font=('Times, Times, Times',8pt);
run;
```

Note: Fonts styles and sizes may be changed to customize the look of the table in the FONT= part of the PROC TEMPLATE code. Additional font choices include Arial, Courier, and others.

3. **Create, within your data step, labels for the variables that you will be using in your table and add a statement to create an 'overall' variable. You will use the 'overall' variable in PROC TABULATE to represent all subjects.**

```
data one;
   set base;

   label male='Gender';
   label black = 'Race';
   label agecat='Age';
   label clbpoutc='Clinic BP';
   label married='Married';
   label clsbp1='Clinic Systolic BP';
   label cldbp1='Clinic Diastolic BP';

   overall=1;
   /* additional statements if needed */
run;
```

4. **Use ODS to: a) open the .rtf file to which you will send your output and b) call your table template using the statement STYLE= janettables ('janettables' is the template created in Step 2 above). You can also give your output a title just below the ODS statement if you want a title presented on every page of your output.**

```
ods rtf style=janettables file= "P:\Projects\Blood Pressure\
sugi08_sas_tables_code.rtf";

title1 "Race and Blood Pressure, Baseline Data";
```

5. **Choose the PROC TABULATE code corresponding to the type of table you want to create from the SAMPLE TABLE CODE section below.**

6. **Clear your title statements (for cleanliness' sake!) and use ODS to close your .rtf output file.**

```
title1;
title2;

ods rtf close;
run;
```

7. **See Appendix for complete program that integrates these steps to produce the tables shown in this paper.**

SAMPLE TABLE CODE

Table 5a. Race and Blood Pressure, Baseline Data
Overall Frequencies

	Overall	
	N	(%)
Race		
Black	306	(50.3)
White	302	(49.7)
Clinic BP		
In Control (< 140/90)	354	(58.2)
Out of Control (>=140/90)	254	(41.8)
Gender		
Female	404	(66.4)
Male	204	(33.6)
Married		
Missing	2	(0.3)
No	300	(49.3)
Yes	306	(50.3)
Age		
25-50	120	(19.7)
51-60	172	(28.3)
61-70	166	(27.3)
>70	148	(24.3)
Missing	2	(0.3)

Table 5a. SAS Code:

```
title2 'Overall Frequencies';

proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male black clbpoutc married agecat overall/style=[cellwidth=3in asis=on];
  tables black clbpoutc male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_.
         agecat agecat_. overall overall.;
run;
```

NOTES:

1. **Do not include the variables that are your column headers in the CLASSLEV statement…doing that will sometimes produce unusual formatting of your table.**
2. **Include ASIS=ON in the CLASSLEV statement to indent the response levels (such as 25-50 and 51-60 under the Age category in the above table). ASIS=ON tells SAS to retain the spaces that you prefixed your response levels with in the VALUE statements in PROC FORMAT (Step 1).**
3. **Use COLPCTN to present column percents (or replace with ROWPCTN for row percents) and F=PCTFMT.. to format your percentages using the PICTURE PCTFMT format you created (Step 2).**
4. **Use MISSTEXT='0' to fill in cells with no counts with a '0'.**

Table 5b. Race and Blood Pressure, Baseline Data
Frequencies, Overall and by BP Control Status

	Overall		BP Control			
			In Control (< 140/90)		Out of Control (>=140/90)	
	N	(%)	N	(%)	N	(%)
Race						
Black	306	(50.3)	156	(44.1)	150	(59.1)
White	302	(49.7)	198	(55.9)	104	(40.9)
Gender						
Female	404	(66.4)	231	(65.3)	173	(68.1)
Male	204	(33.6)	123	(34.7)	81	(31.9)
Married						
Missing	2	(0.3)	1	(0.3)	1	(0.4)
No	300	(49.3)	166	(46.9)	134	(52.8)
Yes	306	(50.3)	187	(52.8)	119	(46.9)
Age						
25-50	120	(19.7)	74	(20.9)	46	(18.1)
51-60	172	(28.3)	106	(29.9)	66	(26.0)
61-70	166	(27.3)	101	(28.5)	65	(25.6)
>70	148	(24.3)	72	(20.3)	76	(29.9)
Missing	2	(0.3)	1	(0.3)	1	(0.4)

Table 5b SAS Code:

```
title2 'Frequencies, Overall and by BP Control Status';

proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male black married agecat/ style=[cellwidth=3in asis=on];
  tables black male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
```

Table 5c. Race and Blood Pressure, Baseline Data
Frequencies, Overall and by Race

			Race			
	Overall		Black		White	
	N	(%)	N	(%)	N	(%)
Clinic BP						
In Control (< 140/90)	354	(58.2)	156	(51.0)	198	(65.6)
Out of Control (>=140/90)	254	(41.8)	150	(49.0)	104	(34.4)
Gender						
Female	404	(66.4)	225	(73.5)	179	(59.3)
Male	204	(33.6)	81	(26.5)	123	(40.7)
Married						
Missing	2	(0.3)	2	(0.7)	0	0
No	300	(49.3)	200	(65.4)	100	(33.1)
Yes	306	(50.3)	104	(34.0)	202	(66.9)
Age						
25-50	120	(19.7)	79	(25.8)	41	(13.6)
51-60	172	(28.3)	93	(30.4)	79	(26.2)
61-70	166	(27.3)	74	(24.2)	92	(30.5)
>70	148	(24.3)	59	(19.3)	89	(29.5)
Missing	2	(0.3)	1	(0.3)	1	(0.3)

Table 5c SAS Code:

```
title2 'Frequencies, Overall and by Race';

proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male clbpoutc married agecat overall/ style=[cellwidth=3in asis=on];
  tables clbpoutc male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         black=  'Race'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
```

Table 5d. Race and Blood Pressure, Baseline Data
Overall and Race-Specific Frequencies, by BP Control Status

	Overall				Black				White			
	BP Control				BP Control				BP Control			
	In Control (< 140/90)		Out of Control (>=140/90)		In Control (< 140/90)		Out of Control (>=140/90)		In Control (< 140/90)		Out of Control (>=140/90)	
	N	(%)	N	(%)	N	(%)	N	(%)	N	(%)	N	(%)
Gender												
Female	231	(65.3)	173	(68.1)	112	(71.8)	113	(75.3)	119	(60.1)	60	(57.7)
Male	123	(34.7)	81	(31.9)	44	(28.2)	37	(24.7)	79	(39.9)	44	(42.3)
Married												
Missing	1	(0.3)	1	(0.4)	1	(0.6)	1	(0.7)	0	0	0	0
No	166	(46.9)	134	(52.8)	105	(67.3)	95	(63.3)	61	(30.8)	39	(37.5)
Yes	187	(52.8)	119	(46.9)	50	(32.1)	54	(36.0)	137	(69.2)	65	(62.5)
Age												
25-50	74	(20.9)	46	(18.1)	46	(29.5)	33	(22.0)	28	(14.1)	13	(12.5)
51-60	106	(29.9)	66	(26.0)	45	(28.8)	48	(32.0)	61	(30.8)	18	(17.3)
61-70	101	(28.5)	65	(25.6)	36	(23.1)	38	(25.3)	65	(32.8)	27	(26.0)
>70	72	(20.3)	76	(29.9)	29	(18.6)	30	(20.0)	43	(21.7)	46	(44.2)
Missing	1	(0.3)	1	(0.4)	0	0	1	(0.7)	1	(0.5)	0	0

Table 5d SAS Code:

```
title2 'Overall and Race-Specific Frequencies, by BP Control Status';
proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male married agecat/ style=[cellwidth=3in asis=on];
  tables male married agecat,
         overall='''*clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         black=  '''*clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         /misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
```

Table 5e. Race and Blood Pressure, Baseline Data
Systolic Blood Pressure Means, Overall

	Overall				
	N	Mean	StdDev	Min	Max
Clinic Systolic BP	608	135.3	20.2	90.0	226.0
Clinic Diastolic BP	608	77.8	11.1	40.0	120.0
Age	606	61.3	12.3	25.0	92.0

Table 5e SAS Code:

```
title2 'Systolic Blood Pressure Means, Overall';
proc tabulate data=one missing order=formatted;
  class overall;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall.;
run;
```

Table 5f. Race and Blood Pressure, Baseline Data
Systolic Blood Pressure Means, Overall and by BP Control Status

						BP Control									
	Overall					In Control (< 140/90)					Out of Control (>=140/90)				
	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max
Clinic Systolic BP	608	135.3	20.2	90.0	226.0	354	122.6	10.9	90.0	139.0	254	153.1	16.5	116.0	226.0
Clinic Diastolic BP	608	77.8	11.1	40.0	120.0	354	74.1	8.4	52.0	89.0	254	83.0	12.3	40.0	120.0
Age	606	61.3	12.3	25.0	92.0	353	60.6	11.6	25.0	89.0	253	62.4	13.0	28.0	92.0

Table 5f SAS Code:

```
title2 'Systolic Blood Pressure Means, Overall and by BP Control Status';
proc tabulate data=one missing order=formatted;
  class overall black clbpoutc ;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1
         max*f=6.1);
  format overall overall. black black_. clbpoutc clbpoutc_. ;
run;
```

Table 5g. Race and Blood Pressure, Baseline Data
Systolic Blood Pressure Means, Overall and by Race

						Race									
	Overall					Black					White				
	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max
Clinic Systolic BP	608	135.3	20.2	90.0	226.0	306	139.5	21.9	90.0	226.0	302	131.1	17.4	90.0	186.0
Clinic Diastolic BP	608	77.8	11.1	40.0	120.0	306	80.1	11.7	52.0	120.0	302	75.5	10.0	40.0	102.0
Age	606	61.3	12.3	25.0	92.0	305	59.3	12.5	28.0	92.0	301	63.4	11.7	25.0	87.0

Table 5g SAS Code:

```
title2 'Systolic Blood Pressure Means, Overall and by Race';
proc tabulate data=one missing order=formatted;
  class overall black;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         black*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall. black black_.;
run;
```

Table 5h. Race and Blood Pressure, Baseline Data.
Overall and Race-Specific Systolic Blood Pressure Means, by BP Control Status

	Overall										Black									
	BP Control										BP Control									
	In Control (< 140/90)					Out of Control (>=140/90)					In Control (< 140/90)					Out of Control (>=140/90)				
	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max
Clinic Systolic BP	354	122.6	10.9	90.0	139.0	254	153.1	16.5	116.0	226.0	156	124.0	10.3	90.0	139.0	150	155.7	18.9	116.0	226.0
Clinic Diastolic BP	354	74.1	8.4	52.0	89.0	254	83.0	12.3	40.0	120.0	156	74.6	8.5	52.0	89.0	150	85.9	11.8	52.0	120.0
Age	353	60.6	11.6	25.0	89.0	253	62.4	13.0	28.0	92.0	156	58.8	12.1	37.0	89.0	149	59.8	12.9	28.0	92.0

	White									
	BP Control									
	In Control (< 140/90)					Out of Control (>=140/90)				
	N	Mean	StdDev	Min	Max	N	Mean	StdDev	Min	Max
Clinic Systolic BP	198	121.5	11.2	90.0	138.0	104	149.5	11.3	130.0	186.0
Clinic Diastolic BP	198	73.8	8.4	52.0	88.0	104	78.7	11.8	40.0	102.0
Age	197	62.1	11.1	25.0	87.0	104	66.0	12.4	29.0	87.0

Table 5h SAS Code:

/*NOTE – The next two statements change the orientation from portrait to landscape within your ODS output document */

```
options orientation=landscape;
ods rtf;

title2 'Overall and Race-Specific Systolic Blood Pressure Means, by BP Control Status';
proc tabulate data=one missing order=formatted;
  class overall black clbpoutc;
  var clsbp1 cldbp1 age;
  tables clsbp1 cldbp1 age,
         overall=''*clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         black=''*clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall. black black_. clbpoutc clbpoutc_. ;
run;
```

/*NOTE – The next two statements change the orientation back from landscape to portrait within your ODS output document */

```
options orientation=portrait;
ods rtf;
```

CONCLUSION

This paper has presented methods that you can use to automate the creation of tables such as demographic tables used in journal articles. It has also described methods/tricks that you can use to format these tables so that they look professional enough for publication. Using these methods requires some upfront coding work, but after the code has been created, its reuse will reduce the amount of time needed to create future tables, especially tables that are produced on a routine basis. The use of these methods can help reduce typos (because the data doesn't have to be manually typed into a table) and the time spent on double-checking numbers that were manually typed in or cut/paste from SAS output into a Microsoft Word document. Finally, the code can serve as a template that can easily be modified to create new styles or table types as the need for new tables arises.

ACKNOWLEDGMENTS

The data used in this presentation are from the Take Control of Your Blood Pressure (TCYB) Study.
NHLBI Grant R01 HL070713 (2003- 2008)
Pfizer Health Communication Initiative Award (2004-2006)
Established Investigator Award, American Heart Association (2006-2011)

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Janet Grubber
VA HSR&D
Hock Plaza, Suite 1105
Box 2720
2424 Erwin Rd.
Durham, NC 27705

Work Phone: 919-668-1489
Fax: 919-668-1300
E-mail: grubb008@mc.duke.edu

APPENDIX – COMPLETE SAS CODE FOR RUNNING PRESENTED TABLES

```
options nofmterr;

proc format;
value yesno_ 1='    Yes'
             0='    No'
             .='    Missing';
value male_ 1='    Male'
            0='    Female';
value agecat_ 1='    25-50'
              2='    51-60'
              3='    61-70'
              4='    >70'
              .='    Missing';
value black_ 1='    Black'
             0='    White';
value clbpoutc_ 1='    Out of Control (>=140/90)'
                0='    In Control (< 140/90)';
value overall 1='Overall';

picture pctfmt(round)low-high= ' 009.9)' (prefix='(');
run;

proc template;
  define style styles.janettables;
    parent=styles.minimal;
    style bodyDate from bodyDate /
           font=('Times',8pt);
    style PagesDate from PagesDate /
           font=('Times',8pt);
    style PagesTitle from PagesTitle /
           font=('Times',8pt);
    style SystemTitle from SystemTitle /
           font=('Times',12pt);
    style Data from Data /
           font=('Times, Times, Times',8pt);
    style Header from HeadersAndFooters /
           font=('Times, Times, Times',8pt);
    style RowHeader from HeadersAndFooters /
           font=('Times, Times, Times',8pt);
  end;
run;

data one;
  set base;

label male='Gender';
label black = 'Race';
label agecat='Age';
label clbpoutc='Clinic BP';
label married='Married';
label clsbp1='Clinic Systolic BP';
label cldbp1='Clinic Diastolic BP';

*creating variable to use in proc tabulate to represent all subjects;
  overall=1;

options orientation=portrait;

ods rtf style=janettables file= "P:\ Projects\Blood Pressure\sugi08_sas_tables_code.rtf";
title1 "Race and Blood Pressure, Baseline Data";
/****************** TABLE 5A in TEXT ************************************************/
title2 'Overall Frequencies';
proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male black clbpoutc married agecat/ style=[cellwidth=3in asis=on];
  tables black clbpoutc male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
/****************** TABLE 5B in TEXT ************************************************/
title2 'Frequencies, Overall and by BP Control Status';
proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
```

```
  classlev male black married agecat/ style=[cellwidth=3in asis=on]; *note - remove column vars
from classlev statement - otherwise get weird results;
  tables black male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=10;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
/****************** TABLE 5C in TEXT ************************************************/
title2 'Frequencies, Overall and by Race';
proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male clbpoutc married agecat/ style=[cellwidth=3in asis=on];
  tables clbpoutc male married agecat,
         overall=''*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         black=  'Race'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
/****************** TABLE 5D in TEXT ************************************************/
title2 'Overall and Race-Specific Frequencies, by BP Control Status';
proc tabulate data=one missing order=formatted;
  class black clbpoutc male married agecat overall;
  classlev male married agecat/ style=[cellwidth=3in asis=on];
  tables male married agecat,
         overall=''*clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)
         black=  ''*clbpoutc='BP Control'*(n*f=4.0 colpctn='(%)'*f=pctfmt.)/misstext='0' rts=15;
  format black black_. male male_. married yesno_. clbpoutc clbpoutc_. agecat agecat_.
         overall overall.;
run;
/****************** TABLE 5E in TEXT ************************************************/
title2 'Systolic Blood Pressure Means, Overall';
proc tabulate data=one missing order=formatted;
  class overall;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall.;
run;
/****************** TABLE 5F in TEXT ************************************************/
title2 'Systolic Blood Pressure Means, Overall and by BP Control Status';
proc tabulate data=one missing order=formatted;
  class overall black clbpoutc ;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall. black black_. clbpoutc clbpoutc_. ;
run;
/****************** TABLE 5G in TEXT ************************************************/
title2 'Systolic Blood Pressure Means, Overall and by Race';
proc tabulate data=one missing order=formatted;
  class overall black;
  var clsbp1 cldbp1 age ;
  tables clsbp1 cldbp1 age,
         overall=''*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         black*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall. black black_.;
run;
/****************** TABLE 5H in TEXT ************************************************/
options orientation=landscape;
ods rtf;

title2 'Overall and Race-Specific Systolic Blood Pressure Means, by BP Control Status';
proc tabulate data=one missing order=formatted;
  class overall black clbpoutc;
  var clsbp1 cldbp1 age;
  tables clsbp1 cldbp1 age,
         overall=''*clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1)
         black=''*clbpoutc='BP Control'*(n*f=4.0 mean*f=6.1 stddev*f=6.1 min*f=6.1 max*f=6.1);
  format overall overall. black black_. clbpoutc clbpoutc_. ;
run;
options orientation=portrait;
ods rtf;
title1;
title2;
ods rtf close;
run;
```